

Transforming co-NP Checks to Answer Set Computation by Meta-Interpretation^{*}

Thomas Eiter and Axel Polleres

Institut für Informationssysteme, TU Wien, A-1040 Wien, Austria
{eiter,polleres}@kr.tuwien.ac.at

Abstract. Many NP-complete problems can be encoded in the answer set semantics of logic programs in a very concise way, where the encoding reflects the typical “guess and check” nature of NP problems: The property is encoded in a way such that polynomial size certificates for it correspond to stable models of a program. However, the problem-solving capacity of full disjunctive logic programs (DLPs) is beyond NP at the second level of the polynomial hierarchy. While problems there also have a “guess and check” structure, an encoding in a DLP is often non-obvious, in particular if the “check” itself is co-NP-complete; usually, such problems are solved by interleaving separate “guess” and “check” programs, where the check is expressed by inconsistency of the check program. We present general transformations of head-cycle free (extended) logic programs into stratified disjunctive logic programs which enable one to integrate such “guess” and “check” programs automatically into a single disjunctive logic program. Our results complement recent results on meta-interpretation in ASP, and extend methods and techniques for a declarative “guess and check” problem solving paradigm through ASP.

1 Introduction

Answer set programming (ASP) [15, 7] is widely proposed as a useful tool for expressing properties in NP, where solutions and polynomial time proofs for such properties correspond to answer sets of normal logic programs, which cover by well-known complexity results the class NP. An example for such a property is whether some given graph has a legal 3-coloring, where any such coloring is itself a certificate for this property.

However, we also might encounter situations in which we want to express a problem which is complementary to some NP problem, and thus belongs to the class co-NP; it is widely believed that in general, not all such problems are in NP and hence not always a polynomial-size certificate checkable in polynomial time exists. One such problem is for instance the property that a graph is *not* 3-colorable. Such properties p can analogously be expressed by a normal logic program (equivalently, by a head-cycle free disjunctive logic program [1]) Π_p , where the property holds iff Π_p has no answer set at all.

Checks in co-NP typically occur as subproblems within more complex problems which have complexity higher than NP, for instance:

^{*} Supported by FWF (Austrian Science Funds) projects P14781 and Z29-N04 and European Commission grants FET-2001-37004 WASP and IST-2001-33570 INFOMIX.

Quantified Boolean Formulas (QBFs): Evaluating a QBF, where we have to check, given a QBF of the form $\exists X \forall Y \Phi(X, Y)$, and an assignment σ to the variables X , whether $\forall Y \Phi(\sigma(X), Y)$ evaluates to true.

Strategic Companies: Checking whether a set of companies is strategic (cf. [9]).

Conformant Planning: Checking whether a given plan is conformant [8], provided executability of actions is polynomially decidable (cf. [4, 18]).

Further examples can be found in [6, 5]. In general, the corresponding logic program Π_p for this check can be easily formulated and the overall problem (evaluating the QBF, finding a strategic companies set resp. a conformant plan) solved in a 2-step approach:

1. Generate a candidate solution by means of a logic program Π_{guess} .
2. Check the solution by another logic program Π_{check} ($=\Pi_p$).

However, it is often not clear how to combine Π_{guess} and Π_{check} into a *single* program Π_{solve} which solves the overall problem. Simply taking the union $\Pi_{guess} \cup \Pi_{check}$ does not work, and rewriting is needed. Theoretical results [6] informally give strong evidence that for problems with Σ_2^P -complexity, it is required that Π_{check} (given as a normal logic program or a head-cycle free disjunctive logic program) is rewritten into a disjunctive logic program Π'_{check} such that the answer sets of $\Pi_{solve} = \Pi_{guess} \cup \Pi'_{check}$ yield the solutions of the problem, where Π'_{check} emulates the inconsistency check for Π'_{check} as a minimal model check, which is co-NP-complete for disjunctive programs. This becomes even more complicated by the fact that Π'_{check} must not crucially rely on the use of negation, since it is essentially determined by the Π_{guess} part. These difficulties can make rewriting Π_{check} to Π'_{check} a formidable and challenging task.

In this paper, we present a generic method for rewriting Π_{check} automatically by using a meta-interpreter approach. Our main contributions are:

(1) We provide a polynomial-time transformation $tr(\Pi)$ from propositional head-cycle-free [1] (extended) disjunctive logic programs (HDLPs) Π to disjunctive logic programs (DLPs), such that the following conditions hold:

T1 Each answer set S' of $tr(\Pi)$ corresponds to an answer set S of Π , such that $S = \{l \mid \text{inS}(l) \in S'\}$ for some predicate $\text{inS}(\cdot)$.

T2 If the original program has no answer sets, then $tr(\Pi)$ has exactly one designated answer set Ω , which is easily recognizable.

T3 The transformation is of the form $tr(\Pi) = F(\Pi) \cup \Pi_{meta}$, where $F(\Pi)$ is a factual representation of Π and Π_{meta} is a fixed meta-interpreter.

T4 $tr(\Pi)$ is *modular* (at the syntactic level), i.e., $tr(\Pi) = \bigcup_{r \in \Pi} tr(r)$ holds. Moreover, $tr(\Pi)$ returns a stratified DLP [16, 17] which uses negation only in its “deterministic” part.

We also describe optimizations and a transformation to positive DLPs, and show that in a precise sense, modular transformations to such programs do not exist.

(2) We show how to use $tr(\cdot)$ for integrating separate guess and check programs Π_{guess} and Π_{check} , respectively, into a single DLP Π_{solve} such that the answer sets of Π_{solve} yield the solutions of the overall problem.

(3) We demonstrate the method on the examples of QBFs and conformant planning [8] under fixed polynomial plan length (cf. [4, 18]), where our method proves to loosen some restrictions of previous encodings.

Our results contribute to deepen the understanding of the guess and check programming paradigm for ASP, and fill a gap by providing an automated construction for integrating guess and check programs. Note that integrated encodings may be straight subject to automated program optimization, which considers both the guess and check part as well as their interaction; this is not possible for separate programs. Furthermore, our results complement recent results about meta-interpretation techniques in ASP, cf. [12, 2, 3].

2 Preliminaries

We assume that the reader is familiar with logic programming and answer set semantics (see [7, 15]) and only briefly recall the necessary concepts.

A *literal* is an atom $a(t_1, \dots, t_n)$, or its negation $\neg a(t_1, \dots, t_n)$, where “ \neg ” (alias, “ $-$ ”) is the strong negation symbol, in a function-free first-order language with at least one constant, which is customarily given by the programs considered. By $|a| = |\neg a| = a$ we denote the atom of a literal. Extended disjunctive logic programs (EDLPs; or simply programs) are finite sets Π of rules r

$$h_1 \vee \dots \vee h_l :- b_1, \dots, b_m, \text{ not } b_{m+1}, \dots, \text{ not } b_n. \quad (1)$$

$l, m, n \geq 0$, where each h_i and b_j is a literal and **not** is weak negation (negation as failure). By $H(r) = \{h_1, \dots, h_l\}$, $B^+(r) = \{b_1, \dots, b_m\}$, $B^-(r) = \{b_{m+1}, \dots, b_n\}$, and $B(r) = B^+(r) \cup B^-(r)$ we denote the head and (pos., resp. neg.) body of rule r . Rules with $|H(r)|=1$ and $B(r)=\emptyset$ are *facts* and rules with $H(r)=\emptyset$ *constraints*. We omit “extended” in what follows and refer to EDLPs as DLPs etc.

Literals (resp. rules, programs) are *ground* if they are variable-free. Non-ground rules (resp. programs) amount to their *ground instantiation*, i.e., all rules obtained by substituting variables with constants from the (implicit) language.

Rules (resp. programs) are *positive*, if “**not**” does not occur in them and *normal*, if $|H(r)| \leq 1$. A ground program Π is *head-cycle free* [1], if no literals $l \neq l'$ occurring in the same rule head mutually depend on each other by positive recursion; Π is stratified [16, 17], if no literal l depends by recursion through negation on itself (counting disjunction as positive recursion).

Recall that the *answer set semantics* [7] for DLPs is as follows. Denote by $Lit(\Pi)$ the set of all ground literals for a program Π . Then, S is an *answer set* of Π , if S is a minimal (under \subseteq) consistent¹ set $S \subseteq Lit(\Pi)$ satisfying all rules in the reduct Π^S , which contains all rules $h_1 \vee \dots \vee h_l :- b_1, \dots, b_m$ for all ground instances of rules (1) in Π such that $S \cap B^-(r) = \emptyset$.

3 Meta-Interpreter Transformation

As mentioned above, a rewriting of a given program Π_{check} to a program Π'_{check} for integrating a guess and a check part into a single program is tricky in general. The working of the answer set semantics is not easy to be emulated in Π'_{check} , since essentially we lack negation in Π'_{check} : Upon a “guess” S for an answer set

¹ For our concerns, we disregard a possible inconsistent answer set.

of $\Pi_{solve} = \Pi_{guess} \cup \Pi'_{check}$, the reduct Π_{solve}^S is **not**-free. Contrary to Π_{check} , there is no possibility to consider varying guesses for the value of negated atoms in Π'_{check} in combination with one guess for the negated atoms in Π_{guess} – all we have is a one in one combination. And, if there is no disjunction in Π'_{check} then Π_{solve} is Horn; thus, its answer sets can be guessed and checked in NP.

This leads us to consider an approach in which the program Π'_{check} is constructed by the use of meta-interpretation techniques [12, 2, 3]: the idea is that a program Π is represented by a set of facts, $F(\Pi)$, which is input to a fixed program Π_{meta} , the meta-interpreter, such that the answer sets of $\Pi_{meta} \cup F(\Pi)$ correspond to the answer sets of Π . Note that existing meta-interpreters are normal logic programs, and can not be used for our purposes for the reasons explained above; we have to construct a novel meta-interpreter which is essentially **not**-free and contains disjunction. To this end, we exploit the following characterization of (consistent) answer sets for HDLPs by Ben-Eliyahu and Dechter [1]:

Theorem 1. *For any ground HDLP Π , a consistent $S \subseteq Lit(\Pi)$ is an answer set iff (1) S satisfies Π and (2) there is a function $\phi : Lit(\Pi) \rightarrow \mathbb{N}$ such that for each literal $l \in S$, there is a rule $r \in \Pi$ with (a) $B^+(r) \subseteq S$, (b) $B^-(r) \cap S = \emptyset$, (c) $l \in H(r)$, (d) $S \cap (H(r) \setminus \{l\}) = \emptyset$, (e) $\phi(l') < \phi(l)$ for each $l' \in B^+(r)$.*

Theorem 1 will now serve as a basis for a transformation from a given HDLP Π to a DLP $tr(\Pi) = F(\Pi) \cup \Pi_{meta}$ such that $tr(\Pi)$ fulfills the properties **T1–T4**:

Input representation $F(\Pi)$ As input for the meta-interpreter Π_{meta} below, we choose the following representation $F(\Pi)$ of the propositional program Π .

We assume that each rule r has a unique name $n(r)$; for convenience, we identify r with $n(r)$. For any rule $r \in \Pi$, we set up in $F(\Pi)$ the facts

$$\begin{array}{ll} \text{lit}(h,l,r) . \quad \text{atom}(l,|l|) . & \text{for each literal } l \in H(r), \\ \text{lit}(p,l,r) . & \text{for each literal } l \in B^+(r), \\ \text{lit}(n,l,r) . & \text{for each literal } l \in B^-(r). \end{array}$$

While the facts for predicate `lit` obviously encode the rules of Π , the facts for predicate `atom` indicate whether a literal is classically positive or negative. We only need this information for head literals; this will be further explained below.

Meta-Interpreter Π_{meta} We construct our meta-interpreter program Π_{meta} , which in essence is a positive disjunctive program, in a sequence of several steps. They center around checking whether a guess for an answer set $S \subseteq Lit(\Pi)$, encoded by a predicate `inS`(\cdot), is an answer set of Π by testing the criteria of Theorem 1. The steps of the transformation cast the conditions of the theorem into rules of Π_{meta} , and provide auxiliary machinery for this aim.

Step 1 We add the following preprocessing rules:

$$\begin{array}{l} 1: \quad \text{rule}(L,R) :- \text{lit}(h,L,R), \text{not lit}(p,L,R), \text{not lit}(n,L,R). \\ 2: \quad \text{ruleBefore}(L,R) :- \text{rule}(L,R), \text{rule}(L,R1), R1 < R. \\ 3: \quad \text{ruleAfter}(L,R) :- \text{rule}(L,R), \text{rule}(L,R1), R < R1. \\ 4: \quad \text{ruleBetween}(L,R1,R2) :- \text{rule}(L,R1), \text{rule}(L,R2), \text{rule}(L,R3), \\ \quad \quad \quad R1 < R3, R3 < R2. \end{array}$$

```

5:   firstRule(L,R) :- rule(L,R), not ruleBefore(L,R).
6:   lastRule(L,R)  :- rule(L,R), not ruleAfter(L,R).
7:   nextRule(L,R1,R2) :- rule(L,R1), rule(L,R2), R1 < R2,
                               not ruleBetween(L,R1,R2).

8:   before(HPN,L,R) :- lit(HPN,L,R), lit(HPN,L1,R), L1 < L.
9:   after(HPN,L,R)  :- lit(HPN,L,R), lit(HPN,L1,R), L < L1.
10:  between(HPN,L,L2,R) :- lit(HPN,L,R), lit(HPN,L1,R),
                               lit(HPN,L2,R), L < L1, L1 < L2.
11:  next(HPN,L,L1,R) :- lit(HPN,L,R), lit(HPN,L1,R), L < L1,
                               not between(HPN,L,L1,R).
12:  first(HPN,L,R)  :- lit(HPN,L,R), not before(HPN,L,R).
13:  last(HPN,L,R)   :- lit(HPN,L,R), not after(HPN,L,R).
14:  hlit(L)         :- rule(L,R).

```

Lines 1 to 7 fix an enumeration of the rules in Π from which a literal l may be derived, assuming a given order $<$ on rule names (e.g. in DLV, built-in lexicographic order; $<$ can also be easily generated using guessing rules). Note that under answer set semantics, we need only to consider rules where the literal l to prove does not occur in the body.

Next, lines 8 to 13 fix enumerations of $H(r)$, $B^+(r)$ and $B^-(r)$ for each rule. The final line 14 collects all literals that can be derived from rule heads. Note that the rules on lines 1-14 plus $F(\Pi)$ form a stratified program, which has a single answer set (cf. [16,17]).

Step 2 We add rules which “guess” a candidate answer set $S \subseteq Lit(\Pi)$ and a total ordering ϕ on S corresponding with the function ϕ in Theorem 1.(2).

```

15:   inS(L) v ninS(L) :- hlit(L).
16:   ninS(L) :- lit(pn,L,R), not hlit(L).           } for each pn ∈ {p,n}
17:   notok :- inS(L), inS(NL), L !=NL, atom(L,A), atom(NL,A).
18:   phi(L,L1) v phi(L1,L) :- inS(L), inS(L1), L < L1.
19:   phi(L,L2) :- phi(L,L1), phi(L1,L2).

```

Line 15 focuses the guess of S to literals occurring in some relevant rule head in Π ; other literals can not belong to S (line 16). Line 17 then checks whether S is consistent, deriving a new distinct atom $notok$ otherwise. Line 18 guesses a strict total order ϕ on inS where line 19 guarantees transitivity; note that minimality of answer sets prevents that ϕ is cyclic, i.e., that $\phi(L,L)$ holds.

In the subsequent steps, we check whether S and ϕ violate the conditions of Theorem 1 by deriving the distinct atom $notok$ in case, indicating that S is not an answer set or ϕ does not represent a proper function ϕ .

Step 3 Corresponding to condition 1 in Theorem 1, $notok$ is derived whenever there is an unsatisfied rule by the following program part:

```

20:   allInSUpto(p,Min,R) :- inS(Min), first(p,Min,R).
21:   allInSUpto(p,L1,R)  :- inS(L1), allInSUpto(p,L,R), next(p,L,L1,R).
22:   allInS(p,R)        :- allInSUpto(p,Max,R), last(p,Max,R).
23:   allNinSUpto(hn,Min,R) :- ninS(Min), first(hn,Min,R).
24:   allNinSUpto(hn,L1,R) :- ninS(L1), allNinSUpto(hn,L,R),
                               next(hn,L,L1,R).
25:   allNinS(hn,R)       :- allNinSUpto(hn,Max,R), last(hn,Max,R).

```

} for each
hn ∈ {h,n}

```

26:  hasHead(R) :- lit(h,L,R).
27:  hasPBody(R) :- lit(p,L,R).
28:  hasNBody(R) :- lit(n,L,R).
29:  allNinS(h,R) :- lit(HPN,L,R), not hasHead(R).
30:  allInS(p,R) :- lit(HPN,L,R), not hasPBody(R).
31:  allNinS(n,R) :- lit(HPN,L,R), not hasNBody(R).
32:  notok :- allNinS(h,R), allInS(p,R), allNinS(n,R), lit(HPN,L,R).

```

These rules compute by iteration over $B^+(r)$ (resp. $H(r)$, $B^-(r)$) for each rule r , whether for all positive body (resp. head and weakly negated body) literals in rule r `inS` holds (resp. `ninS` holds) (lines 20 to 25). Here, empty heads (resp. bodies) are interpreted as unsatisfied (resp. satisfied), cf. lines 26 to 31. The final rule 32 fires exactly if one of the original rules from Π is unsatisfied.

Step 4 We derive `notok` whenever there is a literal $l \in S$ which is not provable by any rule r wrt. `phi`. This corresponds to checking condition 2 from Theorem 1.

```

33:  failsToProve(L,R) :- rule(L,R), lit(p,L1,R), ninS(L1).
34:  failsToProve(L,R) :- rule(L,R), lit(n,L1,R), inS(L1).
35:  failsToProve(L,R) :- rule(L,R), rule(L1,R), inS(L1), L1 != L, inS(L).
36:  failsToProve(L,R) :- rule(L,R), lit(p,L1,R), phi(L1,L).
37:  allFailUpto(L,R) :- failsToProve(L,R), firstRule(L,R).
38:  allFailUpto(L,R1) :- failsToProve(L,R1), allFailUpto(L,R),
                        nextRule(L,R,R1).
39:  notok :- allFailUpto(L,R), lastRule(L,R), inS(L).

```

Lines 33 and 34 check whether condition 2.(a) or (b) are violated, i.e. some rule can only prove a literal if its body is satisfied. Condition 2.(d) is checked in line 35, i.e. r fails to prove l if there is some $l' \neq l$ such that $l' \in H(r) \cap S$. Violations of condition 2.(e) are checked in line 36. Finally, lines 37 to 39 derive `notok` if all rules fail to prove some literal $l \in S$ by iterating over all rules with $l \in H(r)$ using the order from Step 1. Thus, condition 2.(c) is implicitly checked.

Step 5 Whenever `notok` is derived, indicating a wrong guess, then we apply a saturation technique as in [6, 10] to some other predicates, such that a canonical set Ω results. This set turns out to be an answer set iff no guess for S and ϕ works out, i.e., Π has no answer set. In particular, we saturate the predicates `inS`, `ninS`, and `phi` by the following rules:

```

40:  phi(L,L1) :- notok, hlit(L), hlit(L1).
41:  inS(L)    :- notok, hlit(L).
42:  ninS(L)   :- notok, hlit(L).

```

Intuitively, by these rules, any answer set containing `notok` is “blown up” to an answer set Ω containing all possible guesses for `inS`, `ninS`, and `phi`.

3.1 Answer Set Correspondence

Let $tr(\Pi) = F(\Pi) \cup \Pi_{meta}$, where $F(\Pi)$ and Π_{meta} are the input representation and meta-interpreter as defined above. Clearly, $tr(\Pi)$ satisfies property **T3**, and as easily checked, $tr(\Pi)$ is modular. Moreover, \neg does not occur in $tr(\Pi)$ and `not` only stratified. The latter is not applied to literals depending on disjunction; it thus occurs only in the deterministic part of $tr(\Pi)$, i.e. **T4** holds.

To establish **T1** and **T2**, we define the literal set Ω as follows:

Definition 1. Let Π_{meta}^i be the set of rules in Π_{meta} established in Step $i \in \{1, \dots, 5\}$. For any program Π , let $\Pi_\Omega = F(\Pi) \cup \bigcup_{i \in \{1,3,4,5\}} \Pi_{meta}^i \cup \{\text{notok}\}$. Then, Ω is defined as the answer set of Π_Ω .

The fact that Π_Ω is a stratified normal logic program without \neg and constraints, which as well-known has a single answer set, yields the following lemma.

Lemma 1. Ω is well-defined and uniquely determined by Π .

Theorem 2.² For any given HDLP Π the following holds for $tr(\Pi)$:

1. $tr(\Pi)$ has some answer set, and $S' \subseteq \Omega$ for any answer set S' of $tr(\Pi)$.
2. S is an answer set of $\Pi \Leftrightarrow$ there exists an answer set S' of $tr(\Pi)$ such that $S = \{l \mid \text{inS}(l) \in S'\}$ and $\text{notok} \notin S'$.
3. Π has no answer set $\Leftrightarrow tr(\Pi)$ has the unique answer set Ω .

The following proposition is not difficult to establish.

Proposition 1. Given Π , the transformation $tr(\Pi)$, as well as the ground instantiation of $tr(\Pi)$, is computable in LOGSPACE (thus in polynomial time).

As noticed above, $tr(\Pi)$ uses weak negation only stratified and in a deterministic part of the program; we can easily eliminate it by computing in the transformation the complement of each predicate accessed through **not** and providing it in $F(\Pi)$ as facts; we then obtain a positive program. (The built-in predicates $<$ and $! =$ can be eliminated similarly if desired.) However, this modified transformation is not modular. As shown next, this is not incidentally.

Proposition 2. There is no modular transformation $tr'(\Pi)$ from HDLPs to DLPs satisfying **T1**, **T2** and **T3** such that $tr'(\Pi)$ is a positive program.

Proof. Assuming such a $tr'(\Pi)$ exists, we derive a contradiction. Let $\Pi_1 = \{ \mathbf{a} :- \text{not } \mathbf{b}. \}$ and $\Pi_2 = \Pi_1 \cup \{ \mathbf{b}. \}$. Then, $tr'(\Pi_2)$ has some answer set S_2 . Since $tr'(\cdot)$ is modular, $tr'(\Pi_1) \subseteq tr'(\Pi_2)$ holds and thus S_2 satisfies each rule in $tr'(\Pi_1)$. Hence, S_2 contains some answer set S_1 . By **T1**, $\text{inS}(\mathbf{a}) \in S_1$ must hold, and hence $\text{inS}(\mathbf{a}) \in S_2$. By **T1** again, it follows that Π_2 has an answer set S such that $\mathbf{a} \in S$. But the single answer set of Π_2 is $\{ \mathbf{b} \}$, a contradiction. \square

Prop. 2 remains true if **T1** is generalized such that the answer set S of Π corresponding to S' is given by $S = \{l \mid S' \models \Phi(l)\}$, where $\Phi(x)$ is a monotone query (e.g., computed by a normal positive program without constraints). Moreover, if a successor predicate **next**(X, Y) and predicates **first**(X) and **last**(X) for the constants are available (on a finite universe, resp. the constants in Π and rule names), then the negation of the non-input predicates accessed through **not** can be computed by a positive normal program, since such programs capture polynomial time computability by well-known results on the expressive power of Datalog[14]; thus, negation of input predicates in $F(\Pi)$ is sufficient in this case.

² Due to space constraints, we refer to the upcoming extended version of this paper and <http://www.kr.tuwien.ac.at/staff/axel/guessncheck/> for proofs and encodings.

3.2 Optimizations

Π_{meta} can be modified in several respects. We discuss here some modifications which, though not necessarily shrinking size of the ground instantiation, intuitively prune the search of an answer set solver applied to $tr(\Pi)$.

Give up modularity If we sacrifice modularity (i.e. that $tr(\Pi) = \bigcup_{r \in \Pi} tr(r)$), and allow that Π_{meta} partly depends on the input, then we can circumvent the iterations in Step 3 and part of Step 1 as follows: We substitute Step 3 by rules

$$\text{notok} \text{ :- ninS}(h_1), \dots, \text{ninS}(h_i), \text{inS}(b_1), \dots, \text{inS}(b_m), \text{ninS}(b_{m+1}), \dots, \text{ninS}(b_n). \quad (2)$$

for each rule r in Π of form (1). These rules can be efficiently generated in parallel to $F(\Pi)$. Lines 8 to 13 of Step 1 then can also be dropped.

Forcing the guess of S For any normal rule $r \in \Pi$ with $|H(r)| = 1$ which has a satisfied body, we can force the guess of h : we replace (2) by

$$\text{inS}(h) \text{ :- inS}(b_1), \dots, \text{inS}(b_m), \text{ninS}(b_{m+1}), \dots, \text{ninS}(b_n). \quad (3)$$

In this context, since constraints only serves to “discard” unwanted models but cannot prove any literal, we can ignore them during input generation $F(\Pi)$; rule (2) is sufficient. Note that dropping input representation $\text{lit}(\mathbf{n}, l, c)$. for literals only occurring in the negative body of constraints but nowhere else in Π requires some care Such l can be removed by simple preprocessing, though.

Optimize guess of order We only need to guess and check the order ϕ for literals L, L' if they allow for cyclic dependency, i.e., they appear in the heads of rules within the same strongly connected component of the program wrt. S .³ These dependencies wrt. S are easily computed:

$$\begin{aligned} \text{dep}(L, L1) & \text{ :- lit}(h, L, R), \text{lit}(p, L1, R), \text{inS}(L), \text{inS}(L1). \\ \text{dep}(L, L2) & \text{ :- lit}(h, L, R), \text{lit}(p, L1, R), \text{dep}(L1, L2), \text{inS}(L). \\ \text{cyclic} & \text{ :- dep}(L, L1), \text{dep}(L1, L). \end{aligned}$$

The guessing rules for ϕ (line 29 and 30) are then be replaced by:

$$\begin{aligned} \text{phi}(L, L1) \vee \text{phi}(L, L1) & \text{ :- dep}(L, L1), \text{dep}(L1, L), L < L1, \text{cyclic}. \\ \text{phi}(L, L2) & \text{ :- phi}(L, L1), \text{phi}(L1, L2), \text{cyclic}. \end{aligned}$$

Moreover, we add the new atom `cyclic` also to the body of the rules where `phi` appears (lines 36,40) to check `phi` only if Π has *any* cyclic dependencies wrt. S .

4 Integrating Guess and co-NP Check Programs

A general method for solving NP problems using answer set programming is given by the so called “guess and check” paradigm: First a (possibly disjunctive) program is used to guess a set of candidate solutions, and then rules and constraints are added which eliminate unwanted solutions. DLPs allow for the formulation of such problems in a very intuitive way (e.g. solutions of 3-colorability,

³ Similarly, in [1] $\phi : Lit(\Pi) \rightarrow \{1, \dots, r\}$ is only defined for a range r bound by the longest acyclic path in any strongly connected component of the program.

5 Applications

We now exemplify the use of our transformation for two Σ_2^P -complete problems, which thus involve co-NP-complete solution checking: one is about Quantified Boolean formulas (QBFs) with one quantifier alternation, which are well-studied in Answer Set Programming, and the other about conformant planning [4, 18]. Further examples of such problems can be found e.g. in [6, 5, 9] (and solved similarly). However, note that our method is applicable to *any* checks encoded by inconsistency of a HDLP; co-NP-hardness is not a prerequisite.

5.1 Quantified Boolean Formulas

Given a QBF $F = \exists x_1 \cdots \exists x_m \forall y_1 \cdots \forall y_n \Phi$, where $\Phi = c_1 \vee \cdots \vee c_k$ is a propositional formula over $x_1, \dots, x_m, y_1, \dots, y_n$ in disjunctive normal form, i.e. each $c_i = a_{i,1} \wedge \cdots \wedge a_{i,i_i}$ and $|a_{i,j}| \in \{x_1, \dots, x_m, y_1, \dots, y_n\}$, compute the assignments to the variable x_1, \dots, x_m which witness that F evaluates to true.

Intuitively, this problem can be solved by “guessing and checking” as follows:

(QBF_g) Guess a truth assignment for the variables x_1, \dots, x_m .

(QBF_c) Check whether this assignment satisfies Φ for all assignments of y_1, \dots, y_n .

Both parts can be encoded by very simple HDLPs:

$$\begin{array}{ll} QBF_g : & QBF_c : \\ x_1 \vee \neg x_1 \cdot \dots \cdot x_m \vee \neg x_m \cdot & y_1 \vee \neg y_1 \cdot \dots \cdot y_n \vee \neg y_n \cdot \\ & :- a_{1,1}, \dots, a_{1,1_{i_1}} \cdot \dots \cdot :- a_{k,1}, \dots, a_{k,k_{i_k}} \cdot \end{array}$$

Obviously, for any answer set S of QBF_g , representing an assignment to x_1, \dots, x_m , the program $QBF_c \cup S$ has no answer set thanks to the constraints, iff every assignment for y_1, \dots, y_n satisfies formula Φ then. By the method sketched, we can now automatically generate a single program QBF_{solve} integrating the guess and check programs². Note that the customary (but tricky) saturation technique to solve this problem (cf. [6, 9]) is fully transparent to the non-expert.

5.2 Conformant planning

Loosely speaking, planning is the problem to find a sequence of actions $P = \alpha_1, \alpha_2, \dots, \alpha_n$, a *plan*, which take a system from an initial state s_0 to a state s_n in which a goal (often, given by an atom g) holds, where a state s is described by values of fluents, i.e., predicates which might change over time. *Conformant planning* [8] is concerned with finding a plan P which works under all contingencies that may arise from incomplete information about the initial state and/or nondeterministic action effects, which is in Σ_2^P under certain restrictions, cf. [4, 18]. Hence, the problem can be solved with a guess and (co-NP) check strategy.

As an example, we consider a simplified version of the well-known “*Bomb in the Toilet*” planning problem (cf. [4, 13]): We have been alarmed that a possibly armed bomb is in a lavatory which has a toilet bowl. Possible actions are dunking the bomb into the bowl and flushing the toilet. After just dunking, the bomb may be disarmed or not; only flushing the toilet guarantees that it is really disarmed.

Using the following guess and check programs $Bomb_g$ and $Bomb_c$, respectively, we can compute a plan for having the bomb disarmed by two actions:

```

Bombg :                               Bombc :
% Timestamps:                             % Initial state:
time(0). time(1).                         armed(0) v -armed(0).
%Guess a plan:                             % Frame Axioms:
dunk(T) v -dunk(T) :- time(T).            armed(T1) :- armed(T), not -armed(T1),
flush(T) v -flush(T) :- time(T).                time(T), T1=T+1.
% Forbid concurrent actions:               dunked(T1) :- dunked(T), T1=T+1.
:- flush(T), dunk(T).                       % Effect off dunking:
                                              dunked(T1) :- dunk(T), T1=T+1.
                                              armed(T1) v -armed(T1) :- dunk(T), armed(T),
                                              T1=T+1.
                                              % Effect of flushing:
                                              -armed(T1) :- flush(T), dunked(T), T1=T+1.
                                              % Check whether goal holds in stage 2:
                                              :- not armed(2).

```

Bomb_g guesses all candidate plans $P = \alpha_1, \alpha_2$, using time points for action execution, while *Bomb_c* checks whether any such plan P is conformant for the goal $g = \text{not armed}(2)$. Here, CWA on *armed* is used, i.e., absence of *armed*(t) is viewed as *-armed*(t), which saves a negative frame axiom on *-armed*. The final constraint eliminates a plan execution iff it reaches the goal; thus, *Bomb_c* has no answer set iff the plan P is conformant. The answer set $S = \{\text{time}(0), \text{time}(1), \text{dunk}(0), \text{flush}(1)\}$ of *Bomb_g* corresponds to the (single) conformant plan $P = \text{dunk}, \text{flush}$ for goal *not armed*(2).

By our general method, *Bomb_g* and *Bomb_c* can be integrated automatically into a single program $Bomb_{plan} = Bomb_g \cup Bomb_c$ ². It has a single answer set, corresponding to the single conformant plan $P = \text{dunk}, \text{flush}$ as desired.

Note that our rewriting method is more generally applicable than the encoding for conformant planning proposed by Leone et al. [10] who require that state transitions are specified by a positive constraint-free LP. Our method can still safely be used in presence of negation and constraints, provided action execution always leads to a consistent successor state (cf. [4, 18] for a discussion).

6 Discussion and Conclusion

We presented a method for rewriting a head-cycle free (extended) disjunctive logic program (HDLP) into a stratified constraint-free disjunctive logic program, such that their answer sets correspond and a designated answer set of the latter indicates inconsistency of the former. Moreover, we showed how to use this method for automatically integrating a guess and separate check program for a co-NP property (expressed by inconsistency of an HDLP), into an equivalent single (extended) disjunctive logic program. This reconciles pragmatic problem solving with the genuine “guess and check” approach in Answer Set Programming [9], in case a single program expressing the problem is difficult to write, and relieves the programmer from using tricky saturation techniques. Our approach takes advantage of the full expressive power of disjunctive logic programming: integrated encodings as the ones considered are infeasible in less expressive frameworks such as propositional SAT solving or normal logic programming.

For example, [4] described separate interleaved guess and check programs for conformant planning, which are implemented in the planning system $DLV^{\mathcal{K}}$. The present paper solves integrating these programs into a single program, and moreover provides a basis for incorporating harder problems, e.g. checking plan conformance if deciding action executability is already NP-complete (cf. [18]).

References

1. R. Ben-Eliyahu and R. Dechter. Propositional semantics for disjunctive logic programs. *Annals of Mathematics and Artificial Intelligence*, 12:53–87, 1994.
2. J. Delgrande, T. Schaub, and H. Tompits. plp: A generic compiler for ordered logic programs. *Proc. 6th Int'l Conf. on Logic Programming and Nonmonotonic Reasoning (LPNMR-01)*, pp. 411–415. Springer, 2001.
3. T. Eiter, W. Faber, N. Leone, and G. Pfeifer. Computing preferred answer sets by meta-interpretation in answer set programming. *Theory & Practice of Logic Progr.*, 3(4-5):463–498.
4. T. Eiter, W. Faber, N. Leone, G. Pfeifer, and A. Polleres. A logic programming approach to knowledge-state planning, II: The $DLV^{\mathcal{K}}$ system. *Artif. Intell.*, 144(1-2):157-211, 2003.
5. T. Eiter, M. Fink, G. Sabbatini, and H. Tompits. On properties of update sequences based on causal rejection. *Theory & Practice of Logic Progr.*, 2(6):721–777, 2002.
6. T. Eiter, G. Gottlob, and H. Mannila. Disjunctive Datalog. *ACM Transactions on Database Systems*, 22(3):364–418, 1997.
7. M. Gelfond and V. Lifschitz. Classical negation in logic programs and disjunctive databases. *New Generation Computing*, 9:365–385, 1991.
8. R. Goldman and M. Boddy. Expressive planning and explicit knowledge. *Proc. 3rd Int'l Conf. on AI Planning and Scheduling (AIPS-96)*, pp. 110–117, 1996.
9. N. Leone, G. Pfeifer, W. Faber, T. Eiter, G. Gottlob, C. Koch, C. Mateis, S. Perri, and F. Scarcello. The DLV System for Knowledge Representation and Reasoning. Tech. Rep. INFSYS RR-1843-02-14, Inst. f. Informationssysteme, TU Wien, 2002.
10. N. Leone, R. Rosati, F. Scarcello. Enhancing answer set planning. *Proc. IJCAI-01 Workshop on Planning under Uncertainty & Incomplete Information*, pp. 33–42, 2001.
11. V. Lifschitz and H. Turner. Splitting a logic program. *Proc. 11th Int'l Conf. on Logic Programming (ICLP'94)*, pp. 23–37, 1994.
12. V.W. Marek and J.B. Remmel. On the Expressibility of stable logic programming. *Proc. LPNMR 2001*, pp. 107–120. Springer, 2001.
13. D. McDermott. A critique of pure reason. *Comp. Intell.*, 3:151–237, 1987.
14. C. H. Papadimitriou. A note on the expressive power of Prolog. *Bulletin of the EATCS*, 26:21–23, 1985.
15. A. Proveti and T.C. Son, editors. *Proc. AAAI 2001 Spring Symposium on Answer Set Programming: Towards Efficient and Scalable Knowledge Representation and Reasoning*, Stanford, CA, March 2001. AAAI Press.
16. T. Przymusiński. On the declarative and procedural semantics of logic programs. *Journal of Automated Reasoning*, 5(2):167–205, 1989.
17. T. Przymusiński. Stable semantics for disjunctive programs. *New Generation Computing*, 9:401–424, 1991.
18. H. Turner. Polynomial-length planning spans the Polynomial Hierarchy. *Proc. 8th European Conf. on Artificial Intelligence (JELIA)*, pp. 111–124, 2002.